3D grafy

Základní příkazy 3D grafiky a jejich syntaxe:

`plot3d(`*expression*, *ranges*, *options*`)`

`implicitplot3d(`*equations*, *ranges*, *options*`)`

`plots[spacecurve](`*expression*, *range*, *options*`)`

`display3d(`*list*, *options*`)`

Příklad 1

Křivka zvaná šroubovice je dána parametrickými rovnicemi

$$x = r\cos\omega,\ y = r\sin\omega,\ z = v_0\omega,$$

kde r je poloměr příslušného otáčení, ω je úhel tohoto otáčení a v_0 je parametr šroubového pohybu, tzv. redukovaná výška závitu (posunutí podél osy rotace příslušné otočení o jeden radián).
Zvolte si hodnoty r a v_0 a odpovídající šroubovici zobrazte.

Maple - kompletní kód řešení

```
> restart;
> H:=omega->[r*cos(omega),r*sin(omega),v0*omega];
> r:=5: v0:=0.1:
```

Máme dvě možnosti, buď použijeme příkaz `spacecurve`:

```
> plots[spacecurve](H(t),t=0..4*Pi);
```

nebo příkaz `plot3d`:

```
> plot3d(H(t),t=0..4*Pi,s=0..1,numpoints=10000);
```

V druhém případě však musíme z důvodů syntaxe příkazu `plot3d` (očekává dva parametry) uvést „klamný“ parametr s, který s křivkou nijak nesouvisí.

Příklad 2

Plocha zvaná **Plückerův konoid** je dána rovnicí $z = \dfrac{x^2 - y^2}{x^2 + y^2}$. Zobrazte tuto plochu.

Maple - kompletní kód řešení

Užitím příkazu `plot3d`:

```
> z:=(x,y)->(x^2-y^2)/(x^2+y^2);
> plot3d(z(x,y),x=-10..10,y=-10..10):
```

Užitím příkazu `implicitplot3d`:

```
> plots[implicitplot3d](z=(x^2-y^2)/(x^2+y^2), x=-2..2, y=-2..2,z=-2..2,
grid=[40,40,10]):
```

Poznámka: Pomocí příkazu `?plot3d,options` prozkoumejte, jak můžeme ovlivnit podobu grafu. Vyzkoušejte například volby `caption`, `lightmodel`, `style`.

Příklad 3
Průniková křivka kulové plochy s válcovou plochou, které jsou dány rovnicemi

$$x^2 + y^2 + z^2 = 4r, \ (x - r)^2 + y^2 = r^2,$$

se nazývá **Vivianiho křivka** nebo též **Vivianiho okno** (viz Obr.1).



Obrázek 1: Vivianiho okno

Zobrazte příslušnou kulovou a válcovou plochu pro konkrétní hodnotu r. Průnikovou křivku těchto dvou ploch vyjádřete parametricky a zobrazte ve stejném obrázku. Výsledek by měl zhruba odpovídat obrázku 1.

Maple - kompletní kód řešení

```
> restart;
> Sph:=x^2+y^2+z^2=4*r^2; Cyl:=(x-r)^2+y^2=r^2;
> r:=5:
> SphG:=plots[implicitplot3d](Sph,x=-10..10,y=-10..10,z=-10..10, color=grey):
> CylG:=plots[implicitplot3d](Cyl,x=-10..10,y=-10..10,z=-10..10, color=pink):
> plots[display3d](SphG,CylG);
> Sol:=allvalues(solve({Cyl,Sph},{x,y}));
> VW1:=eval([x,y,z],Sol[1]); VW2:=eval([x,y,z],Sol[2]);
> VWG:=plots[spacecurve]({VW1,VW2 },z=-10..10,color=red,thickness=3,numpoints=1000):
> plots[display3d](SphG,CylG,VWG):
```

Chcete-li dostat úplně stejný obrázek, jako je Obr.1, musíte příslušné plochy zadat parametricky:

```
> SphG2:=plot3d([2*r*cos(t)*cos(u),2*r*cos(t)*sin(u),2*r*sin(t)],t=-Pi..Pi,
u=0..2*Pi,scaling=constrained):
> CylG2:=plot3d([r*cos(u)+r,r*sin(u),t],t=-3*r..3*r, u=-Pi..Pi,
scaling=constrained):
> plots[display3d](SphG2,CylG2,VWG):
```

Problém 1
Zobrazte rotační plochu, která vznikne rotací grafu funkce $f(x) = \sin x; \ x \in \langle -\pi; \pi \rangle$ kolem osy x (y).

Problém 2
Möbiova páska - plocha o jedné straně a jedné hraně - může být chápána jako plocha vytvořená pohybem úsečky v prostoru. Najděte transformaci 3D prostoru závislou na parametru t, při níž se úsečka dané délky pohybuje se změnou parametru t podél Möbiovy pásky. Tuto pásku zobrazte.